
About this document

This document is designed to serve as a reference guide for the Baan Application Function Server (AFS). It covers both architecture of and programming with the AFS. Upon completion of this document, the reader will have the necessary knowledge to modify existing AFS code and to create new AFS code.

Chapter 1, “Introduction,” provides a description of the purpose and scope, and lists the definitions, acronyms, and abbreviations.

Chapter 2, “Application Function Server,” describes the architecture and structure of the Application Function Server.

Chapter 3, “4GL engine primitives contains a detailed discussion of the 4GL engine primitives used by the AFS.

Chapter 4, “Special issues,” describes some special topics, such as multi-occurrence / single-occurrence synchronization, message handling, debugging.

Chapter 5, “Guidelines for 4GL application sessions,” contains a number of guidelines for the developers of 4GL application sessions, such as how to handle messages when the session runs in ‘api mode’.

Appendix A, “ASF-DLL example,” contains an example of an AFS-DLL.

References

Developers Guide for BaanERP BOIs (U7611A US).

BOI Builder Developers Guide (U7194B US).
1 Introduction

Purpose

This document is designed to serve as a reference guide for the Baan Application Function Server (AFS). It covers both architecture of and programming with the AFS. Upon completion of this document the reader will have the necessary knowledge to modify existing AFS code and also to create new AFS code.

In this document the restrictions for Baan 4GL application sessions are described, which must be kept in mind during the development and maintenance of these Baan 4GL application sessions. This means that this document must also be used when building new (standard) applications, even when it is not clear yet that the AFS is going to be used together with those new applications.

Scope

The intended audience for this document is experienced Baan programmers only. No attempt is made to explain Baan 4GL programming or Baan architecture outside of what is required for the AFS.

The document applies both to BAAN IV and BaanERP
 in general. However, the implementation and the use of the AFS can be different. When applicable, these differences are noted in the text.

Definitions, acronyms, and abbreviations 

	Term
	Description

	AFS
	Application Function Server (do not confuse this with the DDC (Distributed Data Collection) Function Servers, which is another type of technology)

	AFS-DLL
	A Baan DLL, which contains the calls to the API-handler for a specific Baan 4GL application session

	API
	Application Program Interface

	API-handler
	The program, which serves outside application to call the functionality of the Baan 4GL application sessions

	BCBE
	Baan Connection Basic Edition; this technology is replaced by Baan OpenWorld

	BOI
	Business Object Interface

	FS
	Function Server; this term is sometimes used when AFS is meant


2 Application Function Server

Introduction

The preferred method of integration between Baan (version IVc or later) and third party products is by means of Business Object Interfaces or BOIs. BOIs provide an Application Program Interface (API) for the Baan business logic. BOIs are executed in the context of BCBE or Baan OpenWorld.

The high level interfaces provided by the BOIs cannot be used directly against the Baan business logic due to the Baan architecture. The business logic is implemented by means of sessions, which are primarily user interface based. Programmatic access to sessions is provided by means of a low level message protocol. This message protocol is encapsulated into a set of function primitives called the Application Function Server (AFS); these functions are located in the API-handler. The BOI code accesses the session business logic by means of the AFS. The public, high level interfaces exposed by the BOIs are internally translated to the low level primitives understood by the AFS which then communicates with the session logic by means of the message protocol.

Usually it is not necessary to create BOIs and AFS-DLLs from scratch. BOIs can be generated with the BOI generator (ccboi0102m000 (for BCBE) or ccboi0104m000 (for Baan OpenWorld)). AFS-DLLs can be created with the AFS-DLL generator (ttstpcreatdll). The use of both of these generators is described in [BOIBUILD]. The BOIs and AFS-DLLs created by the generators can be used directly to solve straightforward integration problems where only one BOI and one AFS-DLL is involved to implement a particular interface function set. Complex integration scenarios where multiple BOIs and AFS-DLLs are involved (for example, Sales Orders) require that extra development be performed in the generated BOI and AFS-DLL code. This extra development cannot be accomplished without detailed knowledge of the architecture and programming constructs involved in both the BOI and AFS-DLL code.

The use of an AFS-DLL is optional; the BOIs can also use the function primitives of the AFS directly. However, the function names generated in the AFS-DLL have logical names (for example, field names in stead of field codes), so typing errors will be caught by the compiler. Otherwise, typing errors will be found at runtime. On the other hand, the AFS-DLL creates an extra layer, which will decrease the performance.

The use of AFS is not restricted to BOIs only. Also, other Baan 3GL or 4GL programs can use the AFS to call other sessions with their business logic to perform certain tasks.

The BOI programming constructs are explained adequately in [BOIDEV], therefore only the details of the AFS are explained here.

Architecture

The following diagram shows the role of the AFS in the BOI architecture. The diagram clearly illustrates that the BOI “talks” to the underlying Baan session by means of the AFS.

Explanation:

( 
BOI/3GL/4GL
The program, which wants to access the Baan session business logic.

( 
AFS-DLL
The DLL, which contains wrapper functions to the AFS function primitives. This DLL is optional.

( 
AFS
The DLL, which contains the implementation of the AFS function primitives. This DLL communicates with the 4GL-Engine through a dedicated BMS protocol to let the session execute the requested functionality.

( 
Field buffer
The field buffer contains the fields (names and values) of the form, which are input for the session. When the first API-call to a specific session is done, these fields are retrieved from the session. These can be table fields, but also other form fields, which are used for input.

The values from the BOI are buffered in the field buffer. When the session needs these values (in the field loop of the session), the 4GL-Engine takes the values from the field buffer and makes them available in the session.

After the session changed the field values (during an update, or browse set), the field buffer is updated and the BOI can retrieve the new values. See Chapter 4, for more information.

( 
4GL-Engine
The engine for each Baan 4GL application session, which performs all form-, field-, event-, maintable handling and so forth This is the same 4GL-Engine, which is used when running the session within the user interface (for example, BW). While the session normally waits for user interactions (for example, filling fields, pressing buttons, and so forth), it now waits for AFS commands. The variable api.mode is set to True. This causes the 4GL-Engine to handle differently. Also, the session logic can use this predefined variable, to have different behavior when called from the AFS.

( 
Session logic
The Baan 4GL application session(s), which are carried out with the AFS. More sessions can exist if subsessions are activated by a session or in the case of multi-occurrence/single-occurrence synchronization (BaanERP only).

Only one instance can be active for a session. The AFS keeps track of the session instances and sends the BMS messages to the process number, which is attached to that session instance internally.

Keep in mind that the session only works for one record at a time. For multi-occurrence forms (type 2 or 3), only one occurrence is used.

Example of using AFS 

function long add.record.to.table(
domain dtseno i.seno,






domain dtkey1 i.key1,






domain dtname i.name,






ref string o.mess())

{


long 

retval


string
dummy.msg(1)


stpapi.put.field("dtfsa1101s000", "dtfsa100.seno", str$(i.seno))
|* 1


stpapi.put.field("dtfsa1101s000", "dtfsa100.key1", i.key1)

|* 2


stpapi.put.field("dtfsa1101s000", "dtfsa100.name", i.name)



retval = stpapi.insert("dtfsa1101s000", true, o.mess)

|* 3


if not retval then


retval = stpapi.recover("dtfsa1101s000", dummy.msg)

|* 4


endif


stpapi.end.session("dtfsa1101s000")



|* 5


return(isspace(o.mess))

}

Explanation (the numbers refer to the comments in the source code of the example):

1 
The first API-call for the session (the first argument) results in the activation of the session and the creation of the field buffer. The sections before.program, main.table.io/before.read and form1/before.form of the session are executed if present. Because it is a put.field function, the value of i.seno is put in the field buffer.

2 
Other put functions on a session already activated only result in an update of the field buffer.

3 
During the insert in the session, the relevant sections are executed (choice.add.set/before.choice and so forth) and the field loop is started (all fields of all forms). For these fields, the field sections are executed (before.field, before.checks, check.input and so forth or the check.property hook in the DAL (BaanERP only)). If an error occurs, this is returned in retval, but the message is also returned in o.mess. Because the do.save flag is set to True, the sections choice.update.db/before.choice and so forth are also executed.

4 
Recover is necessary to stop the update mode of the session if an error occurred.

5 
The session must be ended; choice.end.program is executed.

This example is only given here to get an idea of programming with the AFS. In Chapter 3, where the AFS function primitives are described in detail, more examples are given.

Structure of the AFS-DLL

The structure of the AFS-DLL can best be described by looking at an example of one. Appendix A shows the AFS-DLL for the session tcmcs0145m000. The AFS-DLL is realized as a Baan library named tcmcsf0145m000. Note the “f” in the AFS library name. The standard naming convention for an AFS-DLL is to use the name of the parent session and insert an “f” after the module code.

The AFS-DLL contains functions for setting and retrieving values for fields in the underlying session. It also contains functions to add, modify, and delete records from the session. Functions for browsing are also provided. In addition, miscellaneous functions for setting answers to questions, handling subsessions, and retrieving error messages are provided.

Few important features must be noted. Firstly, all of the functions follow a naming convention that is obvious when the example code is perused. The naming convention must be followed when making changes to the AFS-DLL, as the BOI relies on this (if it is generated). Secondly, all of the AFS-DLL functions are merely a thin shell on the low-level primitives exposed by the 4GL Engine (these primitives in turn encapsulate the message protocol described earlier). For more details, refer to the following chapter, which contains an in-depth explanation of the function primitives. An explanation of the function primitives must suffice to also explain the AFS-DLL functions.

Compilation

To compile programs that use the AFS functions, the library ‘ottstpapihand’ must be linked to the program or code ‘#pragma used dll ottstpapihand’ in the script.

3 4GL engine primitives

Introduction

In most AFS functions where a session is mentioned as a parameter, the session is automatically started if it is not already running. Functions, which require the session to have been started already, return an error if no function is called before that started the session (see Chapter 4). This implies that all subsequent function calls for the same session are sent to the same session instance.

The sessions started by the AFS must be 4GL sessions. 3GL sessions (without a form and a script of type 3GL (Without Std.Prog)) cannot be carried out with the AFS.

Get Field Value from session

SYNTAX

void stpapi.get.field (string session, string field, ref string value, [long element])

ARGUMENTS

session
Name of the session this command is executed on.

field
Name of the field whose value is desired.

value
Upon return, this parameter contains the string representation of the current value of the field specified in field. For fields that contain enum or set values, the string representation of the numeric value is returned, not the text description of this value. Date and UTC fields are not converted, so the Baan internal long values are returned as string. Normal numeric values are also returned as string (for example, -123.45), no conversion based on display formats or language dependent format is performed.

element
Array element whose value is to be returned in the case of arrays or repeating fields.

DESCRIPTION

This returns the current value of a particular field from a specified running Baan session. When the field is available in the field buffer, the value is taken from there, otherwise it is extracted directly from the session. See Chapter 4 for the stpapi.* functions, which cause the field buffer to be updated.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa100.seno", str$(i.seno))

retval = stpapi.find("dtfsa1101s000", error.msg)

if retval = 1 then


stpapi.get.field("dtfsa1101s000", "dtfsa100.name", o.name)

stpapi.get.field("dtfsa1101s000", "balance", o.balance)

endif


Explanation:

In the session dtfsa1101s000 the record with key ‘i.seno’ is searched. The values of the fields ‘dtfsa100.name’ (table field on the form) and ‘balance’ (calculated field on the form) are retrieved from the session.

USAGE NOTES
BaanERP only: If the AFS calls a synchronized session (multi-occurrence / single-occurrence) the fields must be taken from the session where the field resides. This implies that if a field must be fetched from the single-occurrence dialog box, a synchronize call must have been performed (see stpapi.synchronize.dialog()). 

Set Field Value in session

SYNTAX

void stpapi.put.field(string session, string field, string value, [long element])

ARGUMENTS

session
Name of the session this command is executed on.

field
Name of the field whose value is desired.

value
The value of the field specified in field is set to the contents of this parameter. Any necessary type conversion is performed, however, Date and UTC fields must be put in the internal Baan format. Enum fields must be put in the internal byte value. Examples: str$(date.num()), str$(-123.45), str$(etol(tcyesno.yes)).

element
Array element whose value is to be set in the case of arrays or repeating fields.

DESCRIPTION

This sets the current value of a particular field in a specified running Baan session. If the field is available in the field buffer, the value is put there till it is processed by the field loop in the session (see Chapter 4). If the field is not available in the field buffer, the field value is directly sent to the session (with the put.var function).

Note that no field sections in the script are executed during the put.field call, so no validation is performed. These sections are called when all fields are processed for an insert or update call.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1201s000", "seno.f", str$(i.seno))


stpapi.put.field("dtfsa1201s000", "seno.t", str$(i.seno))



stpapi.put.field("dtfsa1201s000", "proc.date", str$(date.num()))

stpapi.put.field("dtfsa1201s000", "do.update", str$(etol(dtyesno.no)))

stpapi.continue.process("dtfsa1201s000", error.msg)

Explanation:

For a processing session, the input fields are sent to the session, and the continue process function of the session is executed.

USAGE NOTES

The case and alignment of the passed string values is not relevant. The AFS automatically converts the value to the correct domain.

BaanERP only: If fields are sent to a single-occurrence session, which is synchronized with a multi-occurrence session, the values must be put after a stpapi.synchronize.dialog() call is issued to the multi-occurrence session, because otherwise the single-occurrence session is activated without a link to the multi-occurrence session.

Insert Record in session

SYNTAX

long stpapi.insert (string session, long do.save, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

do.save
Flag that specifies whether a database commit must take place within this function. If do.save is set to 1, then the update.db choice section of the session will be executed before this function returns. If do.save is set to 0, then update.db is not executed. In this case, a stpapi.save() function must be called afterwards to update the database.

err.mesg
This parameter contains the text of the error message if the function cannot complete normally.

DESCRIPTION

This inserts the current record of the specified session into the database. The values of the fields in the session must be set prior to calling this function.

RETURN VALUES

0 Record not inserted or save failed (err.mesg is filled with the reason)

1 Record inserted (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno)) stpapi.put.field("dtfsa1101s000", "dtfsa101.name", name)



retval1 = stpapi.insert("dtfsa1101s000", true, error.msg)



if not retval1 then








retval2 = stpapi.recover("dtfsa1101s000", recover.msg) 

endif

Explanation:

The fields to be filled in the table are sent to the session and the insert function is called.

USAGE NOTES
If err.mesg is filled, stpapi.recover() must be called before any other stpapi.* commands are issued to this session, or the record can be inserted into the database because the stpapi.end.session() call will perform an update.db action.

The error messages returned can be:

( 
Session not available.

( 
Command disabled; no insert is possible in the current state of the session.

( 
Editable synchronized dialog box not started (BaanERP only).

( 
Any error message from the session.

Only use this function with ‘do.save’ 0 if the calling program must distinguish between errors raised by the insert (for example, check.inputs) and the save (for example, skip.io’s in before.write). Do not use it to buffer inserts, because inserts will not be buffered, as the next stpapi.insert() call will execute an ‘update.db’ for the previously inserted record.

If a record is inserted with a type 3 form, the function stpapi.change.view() must be called to set the correct key field values.

BaanERP only: If a record must be inserted with a synchronized single-occurrence dialog box, the dialog box must be synchronized before the first put.field function is called, because otherwise the field buffer is not present. The stpapi.insert() call must be issued to the multi-occurrence session, but the stpapi.put.field() calls to the single-occurrence session. For multi-occurrence sessions of type 3, make sure you call the stpapi.change.view() before the synchronization.

Update Record in session

SYNTAX

long stpapi.update(string session, long do.save, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

do.save
Flag that specifies whether a database commit must take place within this function. If do.save is set to 1, then the update.db choice section of the session is executed before this function returns. If do.save is set to 0, then update.db is not executed. In this case, a stpapi.save() function must be called afterwards to update the database.

err.mesg
This parameter contains the text of the error message if the function cannot complete normally.

DESCRIPTION

This updates the current record of the specified session. The values of the fields in the session must be set prior to calling this function.

RETURN VALUES

0 Record not updated or save failed (err.mesg is filled with the reason)

1 Record updated (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1101s000", error.msg)

if ret = 1 then



stpapi.put.field("dtfsa1101s000", "dtfsa101.name", name)

retval1 = stpapi.update("dtfsa1101s000", true, error.msg)


if not retval1 then





retval2 = stpapi.recover("dtfsa1101s000", recover.msg)

       endif

endif

Explanation:

The record to be updated is searched in the session and its name field is updated.

USAGE NOTES

A record must be current in the session (for example, by a stpapi.find() call), otherwise the results are unpredictable.

If err.mesg is filled, stpapi.recover() must be called before any other stpapi.* commands are issued to this session, otherwise the record can be updated in the database, because the stpapi.end.session() call performs an update.db action.

The error messages returned can be:

( 
Command disabled; no update is possible in the current state of the session.

( 
Editable synchronized dialog box not started (BaanERP only).

( 
Any error message from the session.

Only use this function with do.save 0 if the calling program must distinguish between errors raised by the update (for example, check.inputs) and the save (for example, skip.io’s in before.rewrite). Do not use it to buffer updates, because updates will not be buffered since the next stpapi.find() call to make another record current will execute an update.db for the previously updated record.

BaanERP only: If a record must be updated with a synchronized single-occurrence dialog box, the dialog box must be synchronized before the first put.field function is called, otherwise the field buffer is not present. The stpapi.update() call must be issued to the multi-occurrence session, but the stpapi.put.field() calls to the single-occurrence session.

Delete Record from session

SYNTAX

long stpapi.delete(string session, long do.save, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

do.save
This option should be true. Do.save = false is not supported for stpapi.delete. The option is in the function for compatibility reasons but the function will act like do.save = true.
err.mesg
This parameter contains the text of the error message if the function cannot complete normally.

DESCRIPTION

This deletes the current record of the specified session in the database.

RETURN VALUES

0 Record not deleted or save failed (err.mesg is filled with the reason)

1 Record deleted (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1101s000", error.msg)

if ret = 1 then



retval1 = stpapi.delete("dtfsa1101s000", true, error.msg)


    if not retval1 then





retval2 = stpapi.recover("dtfsa1101s000", recover.msg)

       endif

endif

Explanation:

The record to be deleted is searched in the session and then deleted.

USAGE NOTES

A record must be current in the session (for example, by a stpapi.find() call), otherwise the results are unpredictable.

If err.mesg is filled, stpapi.recover() must be called before any other stpapi.* commands are issued to this session, otherwise the record can be deleted in the database, as the stpapi.end.session() call will perform an update.db action.

The error messages returned can be:

( 
Command disabled; no update is possible in the current state of the session.

( 
Any error message from the session.


BaanERP only: If a multi-occurrence session is used with a synchronized dialog box, the delete action must be sent to the multi-occurrence session.

Save Session Updates to database

SYNTAX

long stpapi.save(string session, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This executes the choice section update.db of the specified session. Note that the same effect can be achieved by issuing any of the following functions:

( 
stpapi.insert()

( 
stpapi.update()

with the do.save parameter set to 1.

RETURN VALUES

0 Record not saved (err.mesg is filled with the reason)

1 Record saved (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1101s000", error.msg)

if ret = 1 then



       stpapi.put.field("dtfsa1101s000", "dtfsa101.name", name)


    retval1 = stpapi.update("dtfsa1101s000", false, error.msg)


    if retval1 then

               retval2 = stpapi.save("dtfsa1101s000", false, error.msg)
endif


       if not retval1 or not retval2 then     


retval3 = stpapi.recover("dtfsa1101s000", recover.msg)

       endif

endif

Explanation:

The record to be updated is searched in the session and its name field is updated. The update function is called without executing update.db directly. The record is saved to the database with the save function.

USAGE NOTES

If err.mesg is filled, stpapi.recover() must be called before any other stpapi.* commands are issued to this session, otherwise the record can be updated in the database as the stpapi.end.session() call will perform an update.db action.

The error messages returned can be:

( 
Command disabled; no update.db is possible in the current state of the session.

( 
Any error message from the session.

BaanERP only: If a record must be saved with a synchronized single-occurrence dialog box, the stpapi.save() call must be issued to the single-occurrence session.

Use the stpapi.insert(), and stpapi.update() calls with the do.save flag set to True, as this is better for performance. Buffering inserts or updates and saving one time will not work, as subsequent stpapi.* function calls will call update.db implicitly. In case of errors, the calling program does not know whether the update.db of the previous record failed, or that the current record is rejected due to field values.

Recover Session updates

SYNTAX

long stpapi.recover(string session, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This executes the choice section recover.set in the specified session.

RETURN VALUES

0 Record not recovered (err.mesg is filled with the reason)

1 Record recovered (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1101s000", error.msg)

if ret = 1 then



       stpapi.put.field("dtfsa1101s000", "dtfsa101.name", name)


    retval1 = stpapi.update("dtfsa1101s000", false, error.msg)


    if retval1 then

               retval2 = stpapi.save("dtfsa1101s000", false, error.msg)
 

       endif


       if not retval1 or not retval2 then     


retval3 = stpapi.recover("dtfsa1101s000", recover.msg)

       endif

endif

Explanation:

The record to be updated is searched in the session and its name field is updated. The update function is called without executing ‘update.db’ directly. The record is saved to the database using the save function.

USAGE NOTES

If err.mesg is filled then stpapi.end.session() must be called before you issue any other commands to this session.

The error messages returned can be:

( 
Command disabled; no update is possible in the current state of the session.

( 
Any error message from the session.

BaanERP only: If a record must be recovered using a synchronized single-occurrence dialog box, the stpapi.recover() call must be issued to the single-occurrence session.

Use the stpapi.insert(), stpapi.update(), and stpapi.delete() calls with the do.save flag set to True, as this is better for performance. Buffering inserts or updates and saving one time will give problems in case records are rejected, because subsequent calls will call update.db implicitly. The calling program does not know then which records failed.

Set Current Record for session

SYNTAX

long stpapi.find(string session [, ref string err.mesg])

ARGUMENTS

session
Name of the session this command is executed on.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This finds the record in the session that corresponds with the current values of the session’s key fields and makes it current. The key field values must be set prior to calling this function.

RETURN VALUES

0 No record found (empty table or error occurred and err.mesg is filled)

1 Record found (err.mesg is empty)

2 A record different to the one requested was found (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1101s000", error.msg)

if ret <> 1 then


        message("Record not found")

endif

Explanation:

The key field for the session is put and the record is made current.

USAGE NOTES

The behavior is the same as a find action through BW: if no record exists, no record is shown (return value 0); if the record is found, then it is selected (return value 1); if the record cannot be found, the next record is selected (return value 2), or if there is no record, no records are shown (return value 0).

The error messages returned can be:

( 
Any error message from the session given during the find action.

Mark Current Record for session

SYNTAX

long stpapi.mark(string session [, ref string err.mesg])

ARGUMENTS

session
Name of the session this command is executed on.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This marks the record in the session, which is made current by the stpapi.find() or one of the stpapi.browse.set() functions.

RETURN VALUES

0 Record is not marked (err.mesg is filled)

1 Record marked (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret <> 1 then

        message("Record not found")

else

        ret = stpapi.mark("dtfsa1501m000", error.msg) 

        if ret then

               stpapi.form.command("dtfsa1501m000", 5, "do.something", 

                                  error.msg)

        endif

endif

Explanation:

Because the form command needs a record to be marked, the found record is marked before the form command is called.

USAGE NOTES

A record must be current in the session (for example, by a stpapi.find() call), otherwise the results are unpredictable.

The error messages returned can be:

( 
Any error message from the session given during the mark action.

Only one record can be marked.

Browse Session records

SYNTAX

long stpapi.browse.set(string session, string option [, ref string err.mesg])

ARGUMENTS

session
Name of the session this command is executed on.

option
This parameter indicates what type of browse action is required. The possible values are first.set, next.set, prev.set, and last.set that correspond with the browse actions first record, next record, previous record, and last record respectively.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This executes the specified browse action against the indicated session. The resulting record is set as current.

RETURN VALUES

0 No record found (empty table, no next or previous record or error occurred and err.mesg is filled)

1 Record found (err.mesg is empty)

EXAMPLE

ret = stpapi.browse.set("dtfsa1101s000", "first.set", error.msg)

while ret


     stpapi.get.field("dtfsa1101s000", "dtfsa100.name", o.name)

        stpapi.get.field("dtfsa1101s000", "balance", o.balance)

        rprt_send()

        ret = stpapi.browse.set("dtfsa1101s000", "next.set", error.msg)

endwhile

Explanation:

This piece of code browses through all records of a session and prints the fields.

USAGE NOTES

For next.set and prev.set a record must be current in the session (for example, by a stpapi.find() call), otherwise the results are unpredictable.

The error messages returned can be:

( 
Any error message from the session given during the browse action

Set Current View for session

SYNTAX

long stpapi.change.view(string session [, ref string err.mesg])

ARGUMENTS

session
Name of the session this command is executed on.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This sets the current view for sessions with forms of type 3 (multiple occurrence plus view). The field values of the view fields must be set prior to calling this function.

RETURN VALUES

0 No view found or error occurred and err.mesg is filled

1 View found (err.mesg is empty)

2 Another view found (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.change.view("dtfsa1501m000", error.msg)

if not ret then



       message(error.msg)

endif

Explanation:

The View field in the session is changed to the given key.

USAGE NOTES

The behavior is the same as a find/new group action through BW: if no records exist, then no record is shown (return value 0); if at least one record in the view exists, then the first record in that view is selected (return value 1); if the view cannot be found, then the first record in the next view selected (return value 2), or if there is no next group, then no records are shown (return value 0).

The error messages returned can be:

( 
Any error message from the session given during the browse action.

Before a record can be inserted on a type 3 form, the view must have been changed to the desired key fields.

Browse Session Views

SYNTAX

long stpapi.browse.view(string session, string option [, ref string err.mesg])

ARGUMENTS

session
Name of the session this command is executed on.

option
This parameter indicates what type of browse action is required. The possible values are first.view, next.view, prev.view, and last.view corresponding to the browse actions first view, next view, previous view, and last view respectively.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This executes the specified browse action against the indicated session. The first record of the view being read is set as Current.

RETURN VALUES

0 No view found or error occurred and err.mesg is filled

1 View found (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.change.view("dtfsa1501m000", error.msg)

if ret then



       ret = stpapi.browse.view("dtfsa1501m000", "prev.view", error.msg)

       if ret then

               message("View before " & str$(i.seno) & " found")

endif

Explanation:

A check is performed whether a view is present before a given view.

USAGE NOTES

The error messages returned can be:

( 
Any error message from the session given during the browse action.

Synchronize Multi-occurrence and Single-occurrence sessions
SYNTAX

long stpapi.synchronize.dialog(string session, string mode, ref string err.mesg)

ARGUMENTS

session
Name of the multi-occurrence session this command is executed on.

mode
The mode in which the synchronized dialog box must be started. The possible values are:


Add – the dialog box is started and synchronized in Edit mode, view fields are sent from the multi-occurrence session to the single-occurrence session; use this mode before a stpapi.insert() call


Modify – the dialog box is started and synchronized in Edit mode, use this mode before a stpapi.update() call


display – the dialog box is started in Display mode


“ ” – no dialog box is synchronized (for future use when multi-occurrence and single-occurrence sessions have the same code, focus is set to the multi-occurrence session)

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This function synchronizes a multi-occurrence session with its registered synchronized dialog box. Depending on the mode, the session with the editable window is synchronized or the read-only window.

RETURN VALUES

0 Sessions could not be synchronized (err.mesg is filled)

1 Sessions are synchronized (err.mesg is empty)

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret then



       ret = stpapi.synchronize.dialog("dtfsa1501m000", "modify", error.msg)

       if ret then

               stpapi.put.field("dtfsa1101s000", "dtfsa101.name", new.name) 


            ret = stpapi.update("dtfsa1501m000", true, error.msg)

       endif

endif

Explanation:

The record is searched in the multi-occurrence session. When found, the synchronized dialog box is started, the field changed, and the record updated.

USAGE NOTES

This function is for BaanERP only.

The error messages returned can be:

( 
Command disabled; no insert/update is possible in the current state of the session.

( 
Session has no synchronized dialog box.

This function must be used to get the multi-occurrence session synchronized with its connected single-occurrence dialog box. It must be called before an insert or update call. It must also be called before executing a Form command on the single-occurrence session, because the editable and display variants can have different sets of form commands. For retrieving data from the single-occurrence dialog box, the synchronization must also be executed.

Send Start processing command to session

SYNTAX

void stpapi.continue.process(string session, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This causes the choice option cont.process to be executed in the specified session

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1201s000", "seno.f", str$(i.seno))


stpapi.put.field("dtfsa1201s000", "seno.t", str$(i.seno))



stpapi.put.field("dtfsa1201s000", "proc.date", str$(date.num()))

stpapi.put.field("dtfsa1201s000", "do.update", str$(etol(dtyesno.no)))

stpapi.continue.process("dtfsa1201s000", error.msg)

Explanation:

For a processing session, the input fields are sent to the session, and the continue process function of the session is executed.

USAGE NOTES
This function is for BAAN IV only. For BaanERP, use the stpapi.form.command().

The error messages returned can be:

( 
Any error message from the session.

When a report is printed, the report specifications must have been set by stpapi.set.report().

Set Session Report parameters

SYNTAX

void stpapi.set.report(string session, string reportname, string device, string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

reportname
Valid Baan report code for desired report.

device
Valid Baan device code for desired device.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This selects the report to be printed and the device to be printed to when the stpapi.print.report(), stpapi.continue.process(), or stpapi.form.command() is called.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1401m000", "seno.f", str$(i.seno))


stpapi.put.field("dtfsa1401m000", "seno.t", str$(i.seno))



stpapi.set.report("dtfsa1401m000", "rdtfsa140111000", pr.device, error.msg)

if isspace(error.msg) then

        stpapi.continue.process("dtfsa1401m000", error.msg)

endif

Explanation:

The specified session will print the given report.

USAGE NOTES
The error message that can be returned is for future use. Currently, no errors will be returned by the function. Errors about invalid reports or devices are returned by the subsequent stpapi.print.report(), stpapi.continue.process(), or stpapi.form.command() calls.

Some sessions determine the report to be printed based on the input on the form. When the report code is already known in the session when the report is opened, the value set in the AFS is ignored.

Only one report can be set, so sessions that print more than one report in one execution cannot be executed with the AFS.

To print the report to a file, use a write-file or rewrite file device. To print the data to another file then the default file for the device, you must also set the spool.fileout variable:

stpapi.put.field(“<sessioncode>”, “spool.fileout”, myfile)

Send Print command to session

SYNTAX

void stpapi.print.report(string session, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This causes the choice option print.data to be executed in the specified session.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1401m000", "seno.f", str$(i.seno))


stpapi.put.field("dtfsa1401m000", "seno.t", str$(i.seno))



stpapi.set.report("dtfsa1401m000", "rdtfsa140111000", pr.device, error.msg)

if isspace(error.msg) then

        stpapi.print.report("dtfsa1401m000", error.msg)

endif

Explanation:

For a print session, the input fields are sent to the session, and the print.data function of the session is executed.

USAGE NOTES
This function is for BAAN IV only. For BaanERP, use stpapi.form.command().

The error messages returned can be:

( 
Report not found.

( 
Device not found.

( 
Any error message from the session.

The report specifications must have been set by stpapi.set.report().

End session

SYNTAX

void stpapi.end.session(string session [, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

DESCRIPTION

This ends the specified session.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno)) stpapi.put.field("dtfsa1101s000", "dtfsa101.name", name)



retval1 = stpapi.insert("dtfsa1101s000", true, error.msg)



if not retval1 then






retval2 = stpapi.recover("dtfsa1101s000", recover.msg) 

endif

stpapi.end.session("dtfsa1101s000")

Explanation:

A record is inserted in the session and the session is ended.

USAGE NOTES
The error messages returned can be:

( 
Any error message from the session.

Although sessions are started implicitly (when used for the first time, see Chapter 4), you must end them explicitly.

When the session has activated another session (see also stpapi.handle.subprocess()), the subsession must be ended first, because the main session may be waiting until the subsession ends. In this case the main session cannot accept the protocol message to end.

Execute session user option

SYNTAX

void stpapi.application.option(string session, long form, long option, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

form
Form number on which user option must be executed. User options (the commands on the Special menu) are defined per form.

option
The option number to be executed. User options appear as choice.user.x options in the Baan 4GL code where x is a number that ranges from 0 to 9. The value of x that corresponds to the option that you want to activate is provided as the value of this parameter.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This executes the specified user option in the session.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret = 1 then



       stpapi.application.option("dtfsa1501m000", 2, 3, error.msg)

endif

Explanation:

For the record found, choice user.3 is executed on the 2nd form.

USAGE NOTES
This function is for BAAN IV only. For BaanERP, use stpapi.form.command().

The error messages returned can be:

( 
Any error message from the session.

Execute session zoom option

SYNTAX

void stpapi.zoom.option(string session, long form, string zoom.prog, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

form
Form number on which zoom must be executed. This needs to be specified, as some sessions depend on a particular form being active when the zoom is executed.

zoom.prog
Baan session for the zoom. This must be the same as the zoom session specified for the choice field of form. If the choice specifies a zoom to a menu, then this session must be one of those mentioned in the menu.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This executes a zoom in the specified session to the program given in zoom.prog.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret = 1 then



       stpapi.handle.subproc("dtfsa1501m000", "dtfsa1201m000", "add")

       stpapi.zoom.option("dtfsa1501m000", 2, "dtfsa1201m000", error.msg)

       stpapi.continue.process("dtfsa1201m000", error.msg)

endif

Explanation:

For the record found, the session dtfsa1201m000 is run, which is connected to the Choice field or a menu choice of the menu connected to the Choice field.

USAGE NOTES
This function is for BAAN IV only. For BaanERP, use stpapi.form.command().

The error messages returned can be:

( 
Any error message from the main session (set in before.choice of choice.zoom).

Execute session form command

SYNTAX

void stpapi.form.command(string session, long command.type, string command.prog, ref string err.mesg)

ARGUMENTS

session
Name of the session this command is executed on.

command.type
The type of form command to be executed. The following values apply:


2 - session


5 - function

command.prog
The code of the session or the name of the function to be executed.

err.mesg
This parameter will contain the text of the error message if the function cannot complete normally.

DESCRIPTION

This function causes the specified form command to be executed in the specified session.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret = 1 then



       ret = stpapi.mark("dtfsa1501m000", error.msg)

       if ret = 1 then



              stpapi.form.command("dtfsa1501m000", 5, "calculate.vat", 

                                  error.msg)

       endif

endif

Explanation:

For the record found, the function calculate.vat is carried out.

USAGE NOTES
This function is for BaanERP only. For BAAN IV, use stpapi.continue.process(), stpapi.application.option(), or stpapi.zoom.option().

The error messages returned can be:

( 
Form command not found in session.

( 
Any error message from the session.

Whether an error message is returned in the function call depends on the way it is coded in the session. See Chapter 4 for more information.

The form command is searched in the session. When it is found in one of the forms it is executed.

If a report is printed, the report specifications must have been set by stpapi.set.report().

If the form command is of type Session, a stpapi.handle.subproc() call must first be issued.

Specify actions for subsessions

SYNTAX

void stpapi.handle.subproc(string session, string sub.prog, string action)

ARGUMENTS

session
Name of the session this command is executed on.

sub.prog
Baan session code to which the action specified applies. This must be the session code of a valid subsession of the specified session or the menu code of a menu that can be started by the specified session.

action
The action to be taken when the subsession starts or which menu choice to be activated when the menu is activated.


Actions for subsessions:


kill - Child process is killed as soon as it starts


ignore - The child process is ignored except for the fact that the parent will wait until the child ends


send - All future stpapi.* function calls that use the name of the parent act on the child instead of the parent


add – The child is added to the list of sessions currently under control. The child can thus be independently controlled by issuing stpapi.* function directly against its session name


Menu choice:


Menu choice of the session to be started converted to a string

DESCRIPTION

This sets the action that is taken when the specified subsession is invoked from the specified session or the menu choice to be activated when the menu is invoked. Note that a stpapi.handle.subproc() must also be called for the session activated by the menu choice.

RETURN VALUES

None

EXAMPLE

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno))

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret = 1 then



       stpapi.handle.subproc("dtfsa1501m000", "dtfsa1201m000", "add")

       stpapi.zoom.option("dtfsa1501m000", 1, "dtfsa1201m000", error.msg)

       stpapi.put.field("dtfsa1201m000", "dtfsa101.date", str$(date.num()))

       stpapi.continue.process("dtfsa1201m000", error.msg)

       stpapi.end.session("dtfsa1201m000")

endif

stpapi.end.session("dtfsa1501m000")

Explanation:

For the record found, the session dtfsa1201m000 is run, which is connected to the Choice field or a menu choice of the menu connected to the Choice field. To also send messages to the subsession, the stpapi.handle.subproc() is called.

USAGE NOTES
When the subprocesses are a session without form, then you cannot communicate with the session through the AFS. In such cases, you cannot define the action for the subsessions of this process.

For this situation, you can define the action for the subprocesses of that session in the group session (session where the pid is the same as the gid of this subsession in the processlist). And the intermediate session must be defined with action Ignore.

This function must always be called when a subprocess is started by the session called through the AFS when the subprocess expects user interaction (such as filling fields or pressing buttons).

Get Messages from session

SYNTAX

string stpapi.get.mess.code(string session [, ref string err.mesg])

ARGUMENTS

session
Name of the session this command is executed on. 

err.mesg
Text of the message.

DESCRIPTION

This retrieves the messages generated by the indicated session as a result of an stpapi.* function call. The messages are returned in the opposite order as generated in the session. Messages generated by the AFS or 4GL-Engine are also returned. These messages are generated when the AFS is not used correctly by the programmer (for example, wrong order of function calls).

RETURN VALUES

String that contains code of error message or the empty string if no error message is found or only the error text is filled.

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno)) stpapi.put.field("dtfsa1101s000", "dtfsa101.name", name)



retval1 = stpapi.insert("dtfsa1101s000", true, error.msg)



if not retval1 then









     while true 

            error.code = stpapi.get.mess.code("dtfsa1101s000", true, error.msg)

            if isspace(error.msg) then

                   break

            endif 




rep.message = error.msg

            rprt_send()

     endwhile

     retval2 = stpapi.recover("dtfsa1101s000", recover.msg)

endif

Explanation:

A record is inserted. If an error occurred, all errors raised by the session are retrieved and printed.

USAGE NOTES
See Chapter 4.

Set answers to questions in session

SYNTAX

void stpapi.enum.answer(string session, string question, bset answer)

ARGUMENTS

session
Name of the session this command is executed on.

question
Baan question code. This question code must be valid within the session, that is, the session must ask this question when the user interface is used.

answer
The enum or set answer to be supplied as answer to the question. This must be expressed as the enum value as opposed to the numeric equivalent for example, tcyesno.yes i.s.o. 1.

DESCRIPTION

This sets the answers to questions that occur while the session is executing.

RETURN VALUES

None.

EXAMPLE

stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(i.seno))

stpapi.put.field("dtfsa1101s000", "dtfsa101.name", new.name)

stpapi.enum.answer("dtfsa1101s000", "dtfsa1101a", tcyesno.yes)

ret = stpapi.insert("dtfsa1101s000", true, error.msg)

Explanation:

The session asks whether the user wants to continue if a record is already present with the same name. The default answer in the session is No, but you must continue.

USAGE NOTES
This function must only be used for questions for which the default answer in the session must be overruled.

If the same question is used more than once in the same session, you can define only one answer.

4 Special issues

To start application sessions

The application session is started when the first AFS call for that session is executed. However, there are some functions that give an error message if they are the first function call for a specific session:

stpapi.insert()
One or more stpapi.put.field() calls must have been performed.

stpapi.update()


A record must be current, so a stpapi.find() must have been performed.

stpapi.delete()
id.

stpapi.save()
A stpapi.insert(), stpapi.update() must have been performed.

stpapi.recover()
id.

stpapi.continue.process()
The session parameters must have been filled.

Field buffer

Field loop

During insert and update calls and also continue.process, print.data, and form.command calls on type 4 forms, the field loop of the 4GL-Engine is executed.

All fields on all forms are processed. For these fields, the 4GL-sections are executed. Between the sections before.input and before.checks, the value is extracted from the field buffer when it is filled there (by a stpapi.put.field() call). To know whether the field is put, another array is used; for each field an entry is in this array which indicates whether the field is put. This array is initialized after each stpapi.* function call, which caused the session to use the values (for example, stpapi.insert()).

Field buffer as input

The following AFS functions cause the 4GL-Engine to take the values from the field buffer:

	Function
	Remark

	stpapi.insert()
	

	stpapi.update()
	

	stpapi.find()
	Only key fields

	stpapi.change.view()
	Only view fields

	stpapi.continue.process()
	Only in a form of type 4, for other form types the record must be made current (with a find or browse function)

	stpapi.print.report()
	Idem

	stpapi.application.option()
	Idem

	stpapi.zoom.option()
	Idem

	stpapi.form.command()
	idem


For the form fields, the field loop is executed (see “Field loop” in this chapter). Messages from the field checks or raised during the processing are returned.

Field buffer as output

The following AFS functions cause the 4GL-Engine to update the values in the field buffer:

	Function
	Remark

	stpapi.insert()
	Field buffer is updated with the values written to the database (for example, the determined order number overwrites the input value, which only contained a series number)

	stpapi.update()
	Idem

	stpapi.save()
	Idem

	stpapi.recover()
	Restores the old values

	stpapi.find()
	The field buffer is updated with the values of the record, which is made Current

	stpapi.browse.set()
	Idem

	stpapi.browse.view()
	By default, the first set of the view is returned, but the application session can react differently on the view events; the field buffer is updated with the values of the current record in the session

	stpapi.synchronize.dialog()
	The field buffer of the single-occurrence session is updated (only of Modify and Display) with the values of the current record of the single-occurrence session


Message handling

Introduction

One of the main issues of using the AFS is modifying the Baan database through the business logic of the sessions. Validations performed in the session scripts must also be performed when the session is executed by the AFS. In case of wrong data, error messages are returned through the AFS to the calling program. This paragraph describes this message handling.

Functions

The following Baan 4GL functions can raise messages:

( 
void message( string mess_str(.) [, arg, ...] )

( 
void mess( string messcode(14), long mode [, arg, ...] )

( 
void set.input.error( string messcode(14) [, arg, ...] )

( 
void dal.set.error.message(string mess.or.code [, arg ...])

( 
void skip.io( string mesg(14) [, ...] )

( 
void abort.io( string mesg(14) [, ...] )

All messages given by these functions are returned by the AFS; however, messages raised by mess() or message() are treated as warning, except when they are followed by set.input.error(), skip.io(), abort.io(), or choice.again().

When the functions set.input.error(), skip.io(), and abort.io() are called with an empty argument and without a mess() or message() call before, an error message is generated. The same applies to the input.again() function.

When a choice.again() is called without a mess() or message() call before it, a Command cancelled warning is generated.

An error is returned in the argument of most stpapi.* functions, warnings can only be retrieved by the function stpapi.get.mess.code().

Message array

Messages raised by the session are kept in an array, with a maximum of 20 messages. When more messages are raised, the 21st message will be: more than 20 messages raised by the session, other messages are lost.

The function stpapi.get.mess.code() gives access to this array. The array has two fields: error code and error text. The error code can be empty for messages raised by the function message(). In the first call to this function, the last message is returned, then the previous one, and so forth.

Example of message array:

	Code
	Text

	dtfsas0002
	Price must be filled

	dtfsas0001
	Note: insufficient inventory


In this example, the session first raised a warning (in check.input of the quantity field) with only a mess() function. In the check.input of the Price field set.input.error(dtfsas0002) was called (or mess(dtfsas0002, 1) followed by set.input.error(“”)). The string “Price must be filled” is also returned in the error argument of the stpapi.insert() call.

When the message array only contains the warning dtfsas0001, no string is returned in the error argument of the stpapi.insert().

See the example code in Chapter 3 how to handle the message array.

Before each stpapi.* call, the array is cleaned up, except stpapi.put.field(), stpapi.get.field() and stpapi.get.mess.code().

Generated error messages

In some situations something is wrong, but there is no error message:

( 
set.input.error, skip.io or abort.io are called without a message code and no message or mess function is called before. For skip.io and abort.io, a default message exists in the 4GL-Engine; this one is returned. For set.input.error, no default error message exists, and the error Input cancelled on field …… is returned.

( 
Input.again is called without a preceding message or mess function call, and the error Input cancelled on field …… is returned.

( 
Choice.again is called without an error message set before. As the AFS does not know whether this function is called before or after the processing, no error message is generated. A warning Command cancelled is written to the message array.

Consider the following examples:


choice.cont.process:

choice.cont.process:


on.choice:


on.choice:


processing()


if some.condition then




choice.again()


    choice.again()




else





    processing()

                                          

  endif

In the first example, processing is performed, in the second it is not. The AFS cannot see the difference. In both situations, the warning Command cancelled is written to the message array and the error message argument of the stpapi.continue.process() or stpapi.form.command() remains empty.

Again Choice.again()

When a choice.again() function is used in the application session, and a message is raised before, the error message argument of stpapi.continue.process() and stpapi.form.command() is filled. This will not always mean that a real error occurred.

Consider the following example:


choice.cont.process:


on.choice:


    processing()


    message(“Ready”)


    choice.again()

In this case, the error message argument is filled with Ready.

To handle this correctly in the calling programs, the programmer must know what happens in the session. If it is impossible to know whether the session is executed as desired, the application session must be changed; see Chapter 5 for more details.

Form commands

Form commands of type Function can also raise messages. Mostly, a message is given and the function is ended by a return statement. In this case, the AFS treats this as warnings, and the error message is not returned directly by the stpapi.form.command() call. When the function is ended by a choice.again(), the error message is returned directly by the stpapi.form.command() call.

Messages from AFS and 4GL-Engine

Several messages can be given by the AFS itself, for example, when functions are not called in the right order. Not all of these messages have message codes assigned to them, so the returned string of the function stpapi.get.mess.code() is empty. Because these messages can change, you must no longer parse these strings.

Multi-occurrence / Single-occurrence

BaanERP only.

For most inserts and updates in BaanERP, a combination of a multi-occurrence and single-occurrence sessions are used. The programmer is responsible to get the sessions in a synchronized state. This is not done automatically for better performance and because the AFS cannot determine whether the editable dialog box or the read-only dialog box must be opened.

With the normal interface, synchronizing the multi-occurrence and single-occurrence session is an asynchronous process, so the processes do not wait until the other is synchronized. The AFS however, will wait until the synchronization is ready.

The sequence of the stpapi.* calls is as much as possible the same as the actions performed with the user interface to do actions on a combination of a multi-occurrence and single-occurrence session.

Synchronization is necessary to:

( 
Insert records.

( 
Update records.

( 
Execute a form command of the single-occurrence session.

( 
Retrieve data from the single-occurrence session.

In the next paragraphs, these actions are described in more detail and for each an example is given.

Inserting records

To insert a record, the multi-occurrence session must be started and also the editable single-occurrence dialog box. The values for the new record must be sent to the single-occurrence session. The stpapi.insert() call is sent to the multi-occurrence session, the stpapi.recover() is sent to the single-occurrence session.

To insert records in a type 3 multi-occurrence session (with view), the view must first be set.

Example type 2, multi-occurrence:

ret = stpapi.synchronize.dialog("dtfsa1501m000", "add", error.msg)

if ret then

       stpapi.put.field("dtfsa1101s000", "dtfsa101.seno", str$(new.seno) 

       stpapi.put.field("dtfsa1101s000", "dtfsa101.name", new.name) 

       ret = stpapi.insert("dtfsa1501m000", true, error.msg)

       if not ret then

              ret = stpapi.recover("dtfsa1101s000", error.msg)

       endif

endif

stpapi.end.session("dtfsa1501m000", error.msg)

Explanation:

By calling the stpapi.synchronize.dialog() function, the multi-occurrence and single-occurrence sessions are both started. The fields are filled for the single-occurrence session. The insert is sent to the multi-occurrence session (the Add button is there). If the insert fails, the recover must be performed on the single-occurrence session (where the Revert button resides). The synchronized dialog box is killed when the multi-occurrence session is ended.

Example type 3, multi-occurrence:

stpapi.put.field("dtfsa1502m000", "dtfsa102.seno", str$(i.seno)

ret = stpapi.change.view("dtfsa1502m000", error.msg)

if ret = 1 then

       ret = stpapi.synchronize.dialog("dtfsa1502m000", "add", error.msg)

       if ret then

           stpapi.put.field("dtfsa1102s000", "dtfsa102.pono", str$(new.pono) 

           stpapi.put.field("dtfsa1102s000", "dtfsa102.name", new.name) 

           ret = stpapi.insert("dtfsa1502m000", true, error.msg)

           if not ret then

                ret = stpapi.recover("dtfsa1102s000", error.msg)

       endif

endif

stpapi.end.session("dtfsa1502m000", error.msg)

Explanation:

Almost the same as the previous example. However, a stpapi.change.view() must be called to fill the View field. During the synchronization, the view fields are copied to the single-occurrence session.

Updating records

To update a record, the multi-occurrence session must be started and also the editable single-occurrence dialog box. The values for the new record must be sent to the single-occurrence session. The stpapi.update() call is sent to the multi-occurrence session, the stpapi.recover() is sent to the single-occurrence session.

No difference exists between type 2 and type 3 multi-occurrence sessions, in both situations the record must be searched in the multi-occurrence session.

Example:

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno)

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret = 1 then

       ret = stpapi.synchronize.dialog("dtfsa1501m000", "modify", error.msg)

       if ret then

           stpapi.put.field("dtfsa1101s000", "dtfsa101.name", new.name) 

           ret = stpapi.update("dtfsa1501m000", true, error.msg)

           if not ret then

                ret = stpapi.recover("dtfsa1101s000", error.msg)

       endif

endif

stpapi.end.session("dtfsa1501m000", error.msg)

Explanation:

The record to be modified is searched in the multi-occurrence session. The single-occurrence session is synchronized and the field is updated.

To execute form commands

To execute form commands of a synchronized single-occurrence session, the dialog box must also be synchronized if the correct record is found in the multi-occurrence session. The mode of the synchronized dialog box depends on the availability of the command in edit- and/or read-only mode.

Note

For form commands on the multi-occurrence session, nothing special needs to be done.

Example:

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno)

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret = 1 then

       ret = stpapi.synchronize.dialog("dtfsa1501m000", "modify", error.msg)

       if ret then

           stpapi.form.command("dtfsa1101s000", 5, "some.function", error.msg) 

       endif

endif

stpapi.end.session("dtfsa1501m000", error.msg)

Explanation:

The record for which the form command must be executed is searched in the multi-occurrence session. The single-occurrence session is synchronized and the form command is executed.

Retrieving data from a synchronized dialog box

Data needed, which is not present in the multi-occurrence session must be retrieved from the synchronized dialog box.

Example:

stpapi.put.field("dtfsa1501m000", "dtfsa101.seno", str$(i.seno)

ret = stpapi.find("dtfsa1501m000", error.msg)

if ret = 1 then

       ret = stpapi.synchronize.dialog("dtfsa1501m000", "display", error.msg)

       if ret then

           stpapi.get.field("dtfsa1101s000", 5, "balance", balance) 

       endif

endif

stpapi.end.session("dtfsa1501m000", error.msg)

Explanation:

The record to be modified is searched in the multi-occurrence session. The single-occurrence session is synchronized and the form command is executed.

Debugging

When you test programs that use the AFS, errors can be in different places:

( 
The AFS is used in the wrong way.

( 
The used Baan 4GL application session is not well designed to be used by the AFS (see Chapter 5).

( 
An error can exist in the communication between the AFS and the 4GL-Engine.

Normal debugging of the AFS-using script and/or the Baan 4GL application session is not always possible, or does not give the appropriate information to solve the problem. For this reason, extra logging facilities are present to help the developers to solve problems.

This logging can be activated by setting the environment variable AFSLOG to a non-blank value:

For ba:
ba6.1 -- -set AFSLOG=1 or AFSLOG=1 ba6.1

For bw:
-- -set AFSLOG=1 in the Command field

During the execution of the AFS-using program, the results of the communication between the AFS and the 4GL-Engine are written to a log file afs.log in the home directory of the user.

Example of logging:

LOGGING STARTED

2001-07-17

15->get.fields

15<-get.fields...

16->get.fields

16<-get.fields...

15->syncadd

15<-syncadd^A0

>dtfsa0101s000 put.field:dtfsa001.seno 3334

>dtfsa0101s000 put.field:dtfsa001.name name

>dtfsa0101s000 put.field:dtfsa001.date 730630

>dtfsa0101s000 put.field:dtfsa001.yeno 1

>dtfsa0101s000 put.field:dtfsa001.mult 3334

>dtfsa0101s000 enum.answer dtfsa0001:2

16->enum.answer^Adtfsa0001^A2

16<-enum.answer^A0

>dtfsa0501m000 insert

16->add.set+save

16<-add.set+save^A0^A0

<dtfsa0501m000 

>dtfsa0501m000 end.session

15->end.program

15<-end.program^A0

In this logging, you can see the AFS functions, which are called by the calling program, how they are translated to the internal protocol between the AFS and the 4GL-Engine and what is returned by the 4GL-Engine.

Text management

Text management is not supported by the AFS.

5 Guidelines for 4GL application sessions

Introduction

This chapter gives some guidelines for Baan 4GL programmers, which constructions cannot be used in the program scripts when the session must be run through the AFS.

Predefined variable api.mode

In 4GL scripts, the predefined variable api.mode is present to know whether the session is started by the AFS. If different behavior is needed for sessions that run in API-mode and sessions with a normal user-interface, this variable must be used.

If a subsession is started from the active session, this activated session is automatically also started in api.mode

Messages

As described in Chapter 4, the message handling is very important. Because there is no user-interface, nobody can see what happened exactly in the session. Functions such as set.input.error(), choice.again(), and so forth, which cause the session to cancel the requested action, can only be used if a message is given, or must not be called in api.mode.

Examples:


choice.cont.process:


before.choice:



if reprint = tcyesno.no and delete = tcyesno.no then




choice.again()



      endif

In this example, the AFS will never know, that nothing is done, while a normal user can see that the hourglass disappears immediately. To solve this, a message must be given, in api.mode anyhow:


choice.cont.process:


before.choice:



if reprint = tcyesno.no and delete = tcyesno.no then

         if api.mode then

              mess("dtfsas0001", 1)

              |* No processing option selected

         endif




choice.again()



      endif

There is an exception for choice.again(). Sometimes it is necessary that the usual behavior of a session is bypassed:


choice.zoom:


before.choice:



if dtfsa001.type = dttype.normal then


      zoom.to$("dtfsa1500m000", z.session, "", "", 0)

         choice.again()



      endif

In this situation, the message Command cancelled is generated, so the AFS-user must ignore the message Command cancelled.

Sometimes messages are returned to the AFS, which contains information only. If this message is followed by a choice.again(), the AFS treats this as an error:


choice.cont.process:


on.choice:



do.all.processing()

      mess("dtfsas0002", 1)



choice.again()

Better is to use the following code:


choice.cont.process:


on.choice:



do.all.processing()

      if not api.mode then

          mess("dtfsas0002", 1)

      endif



choice.again()

Choice.again()

Only use choice.again() to stop the current choice section. Choice.again() used in field sections stops the field loop if records are inserted or updated by the AFS, so fields following on the same form and subsequent forms are not filled with the values from the field buffer.

Execute(…)

To help the end users, sometimes a choice is started automatically, for example, starting the add.set when no records are present. If the session runs in api.mode, starting choices automatically can result in undesired results.

Example:


form.1:


init.form:



execute(find.data)

      if filled.occ = = then

          execute(add.set)

      endif

Better is to use the following code:


form.1:


init.form:


   if not api.mode then

   

 execute(find.data)

          if filled.occ = = then

              execute(add.set)

          endif

      endif

Hidden functionality

The AFS cannot reach functionality that is only accessible through zooming on field level. For example, if for a field the zoom-button is pressed, a menu is shown to maintain or display records. If no separate main session exists to maintain these records, no new record can be entered through the AFS.

Commands

If the session has more than one form, keep the form commands (standard and form specific) the same. AFS does not have a notion of current form.

Appendix A AFS-DLL example

The following example is the AFS-DLL for the Maintain Areas (tcmcs0145m000) session in Baan.

|------------------------------------------------------------------------------

| File created by ttstpcreatdll, Date: 17/02/99

| Created for session: tcmcs0145m000

|------------------------------------------------------------------------------

#pragma used dll ottstpapihand

function extern void f0145m000.put.Area( const domain tccreg value )

{

DLLUSAGE


Function to set Area ( tcmcs045.creg ) in session tcmcs0145m000


arg: - value to put in Area

ENDDLLUSAGE


stpapi.put.field( "tcmcs0145m000", "tcmcs045.creg", value )

}

function extern domain tccreg f0145m000.get.Area( )

{

DLLUSAGE


Function to get Area ( tcmcs045.creg ) from session tcmcs0145m000


return: - value of Area

ENDDLLUSAGE


string value(3)


stpapi.get.field( "tcmcs0145m000", "tcmcs045.creg", value )


return( value )

}

function extern void f0145m000.put.Description( const domain tcdsca value )

{

DLLUSAGE


Function to set Description ( tcmcs045.dsca ) in session tcmcs0145m000


arg: - value to put in Description

ENDDLLUSAGE


stpapi.put.field( "tcmcs0145m000", "tcmcs045.dsca", value )

}

function extern domain tcdsca f0145m000.get.Description( )

{

DLLUSAGE


Function to get Description ( tcmcs045.dsca ) from session tcmcs0145m000


return: - value of Description

ENDDLLUSAGE


string value(30)
mb


stpapi.get.field( "tcmcs0145m000", "tcmcs045.dsca", value )


return( value )

}

function extern void f0145m000.end(ref string error )

{

DLLUSAGE


Function to end connection to session tcmcs0145m000

ENDDLLUSAGE


stpapi.end.session( "tcmcs0145m000", error )

}

function extern long f0145m000.insert( long do.save, ref string error )

{

DLLUSAGE


Function to insert a record in session tcmcs0145m000


Fields must be put before calling this function

ENDDLLUSAGE


return( stpapi.insert( "tcmcs0145m000", do.save, error ) )

}

function extern long f0145m000.update( long do.save, ref string error )

{

DLLUSAGE


Function to update a record in session tcmcs0145m000


Record must be made current and fields to be changed before calling


this function

ENDDLLUSAGE


return( stpapi.update( "tcmcs0145m000", do.save, error ) )

}

function extern long f0145m000.delete( long do.save, ref string error )

{

DLLUSAGE


Function to delete a record in session tcmcs0145m000


Record must be made current before calling this function

ENDDLLUSAGE


return( stpapi.delete( "tcmcs0145m000", do.save, error ) )

}

function extern long f0145m000.save( ref string error )

{

DLLUSAGE


Function to save an update/insert/delete in session tcmcs0145m000

ENDDLLUSAGE


return( stpapi.save( "tcmcs0145m000", error ) )

}

function extern long f0145m000.find( ref string error )

{

DLLUSAGE


Function to find a record in session tcmcs0145m000


Search fields must be put before calling this function

ENDDLLUSAGE


return( stpapi.find( "tcmcs0145m000", error ) )

}

function extern long f0145m000.first( ref string error )

{

DLLUSAGE


Function to find the first record in session tcmcs0145m000

ENDDLLUSAGE


return( stpapi.browse.set( "tcmcs0145m000", "first.set", error ) )

}

function extern long f0145m000.next( ref string error )

{

DLLUSAGE


Function to find the next record in session tcmcs0145m000

ENDDLLUSAGE


return( stpapi.browse.set( "tcmcs0145m000", "next.set", error ) )

}

function extern long f0145m000.previous( ref string error )

{

DLLUSAGE


Function to find the previous record in session tcmcs0145m000

ENDDLLUSAGE


return( stpapi.browse.set( "tcmcs0145m000", "prev.set", error ) )

}

function extern long f0145m000.last( ref string error )

{

DLLUSAGE


Function to find the last record in session tcmcs0145m000

ENDDLLUSAGE


return( stpapi.browse.set( "tcmcs0145m000", "last.set", error ) )

}

function extern long f0145m000.recover( ref string error )

{

DLLUSAGE


Function to undo an update/insert/delete in session tcmcs0145m000

ENDDLLUSAGE


return( stpapi.recover( "tcmcs0145m000", error ) )

}

function extern string f0145m000.get.last.message.code( ref string error )

{

DLLUSAGE


Function to get the code of the message which occurred on 


the last insert/update/delete/save/recover action in session tcmcs0145m000

ENDDLLUSAGE


return( stpapi.get.mess.code( "tcmcs0145m000", error ) )

}

function extern void f0145m000.handle.sub.process( const string sub.process, const string action )

{

DLLUSAGE


Function to define an action when a sub process is started.


Possible actions are: add/send/ignore/kill 



add     - add child session to internal structure, 





session dll of child can be used



send    - send all api calls to child instead of parent



ignore  - child process is ignored, parent will wait





until child exits (for background processes)



kill    - child process is killed immediately

ENDDLLUSAGE


stpapi.handle.subproc( "tcmcs0145m000", sub.process, action )

}

function extern void f0145m000.define.enum.answer( const string question, bset answer )

{

DLLUSAGE


Function to define an answer on a question, when the default answer must not be taken.

ENDDLLUSAGE


stpapi.enum.answer( "tcmcs0145m000", question, answer )

}








































































� Because the AFS is related to the Tools version, this document applies to the Tools versions B40c (for BAAN IV), B50b and 7.1a (for BaanERP). For future versions and releases this document may need modifications.


� BaanERP only


� This code is based on the new AFS function syntax. ttstpcreatdll is not yet changed for this new syntax.

